

NAG Library Routine Document

S14AFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

S14AFF returns the value of the k th derivative of the psi function $\psi(z)$ for complex z and $k = 0, 1, \dots, 4$, via the function name.

2 Specification

```
FUNCTION S14AFF (Z, K, IFAIL)
  COMPLEX (KIND=nag_wp) S14AFF
  INTEGER          K, IFAIL
  COMPLEX (KIND=nag_wp) Z
```

3 Description

S14AFF evaluates an approximation to the k th derivative of the psi function $\psi(z)$ given by

$$\psi^{(k)}(z) = \frac{d^k}{dz^k} \psi(z) = \frac{d^k}{dz^k} \left(\frac{d}{dz} \log_e \Gamma(z) \right),$$

where $z = x + iy$ is complex provided $y \neq 0$ and $k = 0, 1, \dots, 4$. If $y = 0$, z is real and thus $\psi^{(k)}(z)$ is singular when $z = 0, -1, -2, \dots$.

Note that $\psi^{(k)}(z)$ is also known as the *polygamma* function. Specifically, $\psi^{(0)}(z)$ is often referred to as the *digamma* function and $\psi^{(1)}(z)$ as the *trigamma* function in the literature. Further details can be found in Abramowitz and Stegun (1972).

S14AFF is based on a modification of the method proposed by K lbig (1972).

To obtain the value of $\psi^{(k)}(z)$ when z is real, S14AEF can be used.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

K lbig K S (1972) Programs for computing the logarithm of the gamma function, and the digamma function, for complex arguments *Comp. Phys. Comm.* **4** 221–226

5 Arguments

1: Z – COMPLEX (KIND=nag_wp) Input

On entry: the argument z of the function.

Constraint: Z must not be ‘too close’ (see Section 6) to a non-positive integer when $Z = 0.0$.

2: K – INTEGER Input

On entry: the function $\psi^{(k)}(z)$ to be evaluated.

Constraint: $0 \leq K \leq 4$.

3: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $K < 0$,
or $K > 4$,
or $\text{Re}(Z)$ is ‘too close’ to a non-positive integer when $\text{Im}(Z) = 0.0$. That is,
 $\text{abs}(\text{Re}(Z) - \text{nint}(\text{Re}(Z))) < \textit{machine precision} \times \text{nint}(\text{abs}(\text{Re}(Z)))$.

IFAIL = 2

The evaluation has been abandoned due to the likelihood of overflow. The result is returned as zero.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Empirical tests have shown that the maximum relative error is a loss of approximately two decimal places of precision.

8 Parallelism and Performance

S14AFF is not threaded in any implementation.

9 Further Comments

None.

10 Example

This example evaluates the psi (trigamma) function $\psi^{(1)}(z)$ at $z = -1.5 + 2.5i$, and prints the results.

10.1 Program Text

```

Program s14affe

!      S14AFF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: nag_wp, s14aff
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Complex (Kind=nag_wp)      :: y, z
Integer                    :: ifail, ioerr, k
!      .. Executable Statements ..
Write (nout,*) 'S14AFF Example Program Results'

!      Skip heading in data file
Read (nin,*)

      Write (nout,*)
      Write (nout,*) '      Z      K      (d^K/dz^K)psi(Z)'
      Write (nout,*)

data: Do
  Read (nin,*,Iostat=ioerr) z, k

  If (ioerr<0) Then
    Exit data
  End If

  ifail = -1
  y = s14aff(z,k,ifail)

  If (ifail<0) Then
    Exit data
  End If

  Write (nout,99999) z, k, y
End Do data

99999 Format (1X,'(',F5.1,',',F5.1,')',I6,'      (',1P,E12.4,',',E12.4,')')
End Program s14affe

```

10.2 Program Data

```

S14AFF Example Program Data
(-1.5, 2.5)  1 : Values of Z and K

```

10.3 Program Results

```

S14AFF Example Program Results

      Z      K      (d^K/dz^K)psi(Z)
( -1.5,  2.5 )  1  ( -1.9737E-01, -2.4271E-01 )

```
